Development of Wave Train Analysis Package
—— WAVETRAP ——

Ryuichiro NISHI*, Michio SATO* and Nicholas C.KRAUS**

Abstract

Ocean waves are composed of a number of waves with different wave heights, periods, frequencies, and directions. This sea state is mainly analyzed by two approaches. One is by spectral analysis, and the other by individual wave analysis. In spectral analysis, it is assumed that the sea is composed of a variety of sinusoidal waves with different amplitudes, frequencies, and directions. Therefore, each wave component can be linearly superimposed to generate the original sea state. On the other hand, individual wave analysis distinguishes and identifies each wave, then calculates the wave height and period of each wave passing through a fixed wave station. Therefore, individual wave analysis is more convenient to study the breaking wave mechanism and related cross-shore sediment transport in the surf zone. This report describes only analysis procedures and algorithms that can be used to determine individual-wave properties of a train of waves with respect to statistics. Thus, the development of a spectral analysis package is left for future study. The analysis concerns properties such as wave height, wave period, water surface elevation, extreme wave properties, joint distribution of height and period. In addition, this program is simple to extend to the analysis of a multi-burst wave record, but it is expected that when using the computer program, engineers will take responsibility themselves for the listings.

1. Introduction

Ocean waves such as shown in Fig.1 are classified either as monochromatic (regular) waves or as random (irregular) waves, based on the shape of the free surface elevation. Monochromatic waves can be characterized by a single amplitude and frequency, where the wave height is twice the amplitude for the case of linear or small-amplitude waves, and the wave period is the reciprocal of the frequency. When long swell generated by a storm approaches to the coast, the clear regularity of waves is seen in space and time as shown in Photograph 1. However, the real sea is normally more complex than a simple monochromatic wave. A real sea is composed of a number of waves with different wave heights, periods, and directions. In the present report on individual wave analysis, wave direction is not considered as a variable, and an



Photograph 1 Regularly swell waves.



Photograph 2 Breaking waves in the surf zone

Received May 31, 1997

* Dept. of Ocean Civil Engineering

** US Army Engineer Waterways Experiment Station, Coastal and Hydraulics Laboratory



Fig. 1 Example of free surface elevation.



(a) wave height distribution (individual wave analysis) (b) Energy spectrum (spectral analysis)

Fig. 2 Character of random waves

analysis is made for a train of waves independent of direction.

As well-known by ocean and coastal engineers, the sea state is mainly analyzed by two approaches. One is by individual wave analysis, and the other by spectral analysis as shown in Fig. 2. In spectral analysis, it is assumed that the sea is composed of a variety of sinusoidal waves with different amplitudes and frequencies, and direction, which is omitted here. Therefore, each wave component can be linearly superimposed to generate the original waves. On the other hand, individual wave analysis distinguishes or identifies each wave and calculates the wave height and period of each wave passing a fixed point which is normally at the wave gauge. Both methods have advantages and disadvantages. For instance, individual wave analysis may be used to calculate the design wave height for a coastal structure and to study the breaking wave mechanism shown in Photograph 2, whereas spectral analysis might be used to estimate the radiation stress in the surf zone.

This report describes analysis procedures and



Fig. 3 Free surface elevation and definition of zero-up and zero-down cross methods.

algorithms that can be used to determine the individual-wave properties of a train of waves. The analysis thus concerns properties such as wave height, wave period, water surface elevation, extreme wave properties, and joint distribution of height and period. Procedures used are those common in coastal engineering and science, and this paper serves to document the analysis procedures used at the Coastal Engineering Research group, Dept. of Ocean Civil Engineering, Faculty of Engineering, Kagoshima University.

Program listings, execution procedure, and calculation results are provided. A short description of individual wave analysis is given in Chapter 2, and the associated computational results are in Chapter 4. Sample program shown in appendix is for single wave record, however the program can be easily extend to the multi-burst wave record.

2. Elements of Individual Wave Analysis

It is assumed that the sea surface is composed of a variety of waves. Each wave can be identified by applying an individual wave algorithm to analyze a record of free-surface elevation. To do this, the record is decomposed into individual waves by either the zero-up cross or zero-down cross method as shown in Fig. 3. A wave is distinguished by two successive zero-up cross points, or by two successive zero-down cross points, respectively. In addition, the mean water level (MWL) is defined as the average of the free surface elevation between two zero-up cross points or two zero-down cross points, viz. one wave average in this study, while many literature refers the mean water level (MWL) as a low-frequency-passed-signal.

If the height of individual waves contained in a deep-water record or at an offshore wave gauge record are identified and ranked from the highest to the lowest, calculation of the highest waves is straightforward. Typically, the maximum wave height which is the design wave height for the important coastal structures, the one-tenth wave height, the one-third wave height, mean wave height, and the root-mean-square (rms) wave height are the widely used statistically representative waves among the possible highest waves. Historically, SIO (1947) defined the significant wave height as the average height of the highest third of the waves in a record and the terminology of the significant wave was given by Wiegel and Saville (1996) in detail.

Assuming that N incident waves are recorded at the wave gauge station, a representative wave height with an occurrence probability, p (=n/N, where n is the number of waves considered in the calculation), can be expressed as H_p. The quantity H_p is normally written as $H_{1/10}$, $H_{1/3}$, or H_{mean} ($=H_{1/1}$). On the other hand, the rms wave height H_{rms}, which express the wave energy is defined as follows;

$$H_{rms} = \sqrt{\frac{1}{N}\sum_{i=1}^{N} H_i^2} \tag{1}$$

In the spectral analysis, the sea surface can be expressed as a linear composition of sinusoidal waves as follows:

$$\eta(t) = \sum_{j=1}^{\infty} a_j \cos(w_j t + \alpha_j) \tag{2}$$

where $\eta(t)$ is a free surface elevation as a function of time, a_j is the amplitude of j^{th} component wave, w_j is the angular frequency of j^{th} component wave, and α_j is the phase angle of the j^{th} component wave at the time $t=0$.

It is noted that the *rms* wave height can be defined in a spectral analysis approach (e.g. Dean and Darlymple, 1991), if wave shape is not severely deformed by shallow water depth or high wave steepness.

$$H_{rms} \approx 2\sqrt{2\sigma} \tag{3}$$

where

$$\sigma^2 = \int_0^\infty E(\omega)d\omega \tag{4}$$

and $E(\omega)$ is the continuous energy spectrum. Therefore, it is noted that there is two definition of H_{rms}.

2.1 Rayleigh Probability Distribution

The wave height is generally found to follow a Rayleigh distribution. The probability density function can be expressed as

$$\begin{aligned} pdf(H) &= \frac{d}{dH}(p(H \le \hat{H}) = \frac{d}{dH}(1-e^{(H/H_{rms})}) \\ &= 2H\frac{e^{-(H/H_{rms})^2}}{H_{rms}^2} \end{aligned} \tag{5}$$

For the Raleigh distribution, the mean wave height is calculated as

$$\begin{aligned} \overline{H} &= \frac{\int_0^\infty H\cdot pdf(H)dH}{\int_0^\infty pdf(H)dH} \\ &= \frac{2}{H_{rms}}\int_0^\infty \frac{H^2 e^{(H/H_{rms})^2}}{H_{rms}^2} d\left(\frac{H}{H_{rms}}\right) \end{aligned} \tag{6}$$

From the statistical approach by Longuet-Higgins (1957), the highest waves H_p can be expressed as:

$$\frac{H_p}{H_{rms}} = \sqrt{\ln\frac{1}{p}} \tag{7}$$

Therefore,

$$\overline{H_p} = \frac{\int_{H_p}^\infty H\cdot pdf(H)dH}{\int_{H_p}^\infty pdf(H)dH} = H_{rms}\frac{\int_{H_p/H_{rms}}^\infty x^2 e^{x^2}dx}{\int_{H/H_{rms}}^\infty x e^{x^2}dx} \tag{8}$$

where, x is a dummy variable. This integration can be done by parts as follows:

$$\begin{aligned} \frac{\overline{H_p}}{H_{rms}} &= \frac{H_p/H_{rms} e^{(H_p/H_{rms})^2} + \int_{H_p/H_{rms}}^\infty e^{x^2}dx}{e^{(H_p/H_{rms})^2}} \\ &= \sqrt{\ln\frac{1}{p}} + \frac{\sqrt{\pi}}{2p} erfc\left(\sqrt{\ln\frac{1}{p}}\right) \end{aligned} \tag{9}$$

where, erfc(x) is the error function. Assuming the Rayleigh wave height distribution, representative wave heights are expressed as follows:

$$\begin{aligned} H_{1/10} &= 1.80H_{rms} \\ H_{1/3} &= 1.416H_{rms} \\ H_{mean} &= 0.886H_{rms} \end{aligned} \tag{10}$$

Many textbooks, for instance Goda (1985), Dean and Dalrymple (1991), can be referred for more information on wave statistics.

2.2 Rayleigh Distribution for Wave Period

Wave height distribution is usually expressed by the Rayleigh distribution. In contrast, a distribution of square of wave period can be denoted by the Rayleigh distribution as follows:

$$p(T)dT = 2.7\frac{T^3}{T^4}exp[-0.675(\frac{T}{T})^4]dT \qquad (7)$$

In addition to the previous two distribution, i.e., wave height and wave period, joint distributions of wave height and period are calculated by the program developed here.

3. Calculation Algorithms

Accurate records of the water surface elevation can be obtained by several types of measurement techniques. For instance, the photopole method is to gauge incident waves by a photographic record or a video record of the water surface elevation at a pole (Hotta and Mizuguchi, 1980). In contrast to this direct measurement, indirect measurements such as from a pressure gauge, capacitance gauge, resistance gauge, current meter, or an acoustic gauge are often used. In general, these sensors can measure time series of surface elevation η (t) and pressure, p(t), where t is the elapsed time, while direct measurement is more time consuming to convert the analog wave data to digital values.

In engineering applications, wave data is probably stored or provided by either as a single record or as a multi-burst record consisting of measurements made at fixed time interval. Therefore, two programs are developed for the present study, one for single record of wave data, and the other for multi-burst records of wave data. The programs to analyze single record is named WAVETRAP (WAVE Train Analysis Program) temporarily. It is emphasized that this program is the first generation. The program list is attached in appendix for single record. Therefore the reader can easily modify the program depending on the engineering project under their own responsibility.

3.1 Execution of WAVETRAP

This program calculates statistical information from individual waves measured at a fixed location. The present version of the program processes water

```
1.   Compile and link the WTES1.FOR.
2.   Run the WTES.EXE.
3.   Wait a few seconds while the program executes.
4.   The output file WSINFO1.DAT will be generated
```

Fig. 4 Execution procedure of the WAVETRAP

```
(1) Print out of "WSINFO1.DAT"

WAVE INFORMATION BY ZERO CROSS METHOD
DATA ANALYSIS BY V970430OCE
WAVE DATA         = PTEST1.DAT
number of data =          1024
sampling time  =     0.500000      sec

r.m.s n(t) =     0.619663
skewnwss    =    -0.279487E-01
kurtosis   =            2.0

zero-up cross method
NUMBER OF WAVES =                  78
Hmax     =      2.98515
H(1/10)=      2.74907
H(1/3) =      2.34486
Hmean   =       1.60095
Tmax     =      6.35514
T(1/10)=      6.55067
T(1/3) =      6.68782
Tmean   =       6.51892

zero-down cross method
NUMBER OF WAVES =                  77
Hmax     =      3.47690
H(1/10)=      2.91883
H(1/3) =      2.32934
Hmean   =       1.59667
Tmax     =      6.29060
T(1/10)=      6.55200
T(1/3) =      6.63559
Tmean   =       6.53182
```

Fig. 5 General wave information produced by the program

surface elevation η (t) taken by either capacitance type or resistance type wave gauges depending upon the application. A subprogram which convert pressure data to surface elevation will be added in the next generation of the program. The execution procedure and output are shown in Fig. 4 and 5 respectively. Whenever a user would like to obtain the statistical information of velocities, u and v, these data can be introduced into a η array instead of η data.

The program produces output data files which are ready for graphic software. A set of data is formatted in either (x, y) or (x, y, z) form. For a user who needs graphical results, see the READMEWS. FOR file.



Fig. 6 Cross-shore distributions of wave heights distribution by zero-up cross method vs. spectral method



Fig. 7 Cross-shore distributions of wave height distribution by zero-down cross method vs. spectral method.



Fig. 8 Time series of wave height, wave period, and mean water level by zero cross method



(a)Probability of wave height by zero cross method



(b)Probability of wave height by zero cross method

Fig. 9 Probability of wave height by

4. Example Calculation Results

The algorithms and programs were examined by comparing calculated results to spectral analysis results for $H_{n,o}$ and H_{rms} data obtained in the SUPERTANK Data Collection Project as published in Smith and Kraus (1995). In general, $H_{1/3}$ and H_{mo} are almost identical in deeper water, and H_{rms} by individual wave analysis and by spectral analysis are also



(a) zero-up cross method

(b) zero-down cross method

Fig. 10 Joint distribution of wave height and period by

similar, but not identical because of the different definitions associated with the two methods. The definitions of these variables are given in the Chapter 2.

The comparison of the wave height distribution obtained at sixteen wave gauges are shown in Figures 6 and 7, for which the application of the zero-up cross method and zero-down cross method are shown, respectively. In general, the computational results by the program are in reasonable agreement compared to the spectral analysis output. For instance, the significant wave height is slightly smaller than the 0 order moment wave height, H_{mo}, in the offshore region and slightly larger in the surf zone. In addition, the root-mean square wave height defined by zero-up cross is slightly smaller than H_{rms} defined by spectral analysis in the offshore region, but slightly higher than that in the surf zone. The additional output of the wave height distribution and the joint distribution of wave height and wave period are also shown in Fig. 8, 9, and 10, respectively.

5. Conclusions

The wave train analysis package named the "WAVETRAP" has been developed to analyze the sea state. The analysis package contains an individual wave analysis method, but the spectral analysis has been left for the further study. It is emphasized that the individual wave analysis is superior to reveal the characteristics of wave train in time domain, but not in the frequency domain. The comparison of calculated wave statistics to the spectral result obtained in the SUPERTANK project showed a good agreement. Therefore, the current program package is allowed to open for public access. The program is coded by the FORTRAN so that an average engineer can modify the code for his/her own purpose.

The current package has been applied to the other laboratory data. The analysis of the large wave tank data showed that the mean wave period decreases especially beyond the breaking point and increases in a swash zone. In other word, the number of waves which pass through each wave gauge stations increases in the surf zone, but decreases in a swash zone. As seen in many literature and reports, the change in wave height in surf zone has been focused to relate to the cross-shore sediment transport, however it is emphasized that the change in wave period in the surf zone and in the swash zone should be considered, because the wave period is a key parameter to calculate the transport rate, current velocity, and shear stress, etc. Again, it is noted that the wave period is not constant across the surf and swash zones.

The aim of this study is to develop an individual wave analysis package and not limited to the application itself to certain project. So, further wave analysis had never been conducted in this study. Finally, it is noted that the wave train analysis package "WAVETRAP" is open for public access, however authors will not take any responsibility to the users application and result. It is also encouraged to include some other subroutines to improve the analysis package.

References

Dean, R. G. and R. A. Darlymple ; Water wave mechanics for engineers and scientists, World Science 1991.

Ebersole, B. A. and Hughes, S. A.; DUCK85 photopole experiment, Miscellaneous paper CERC-87 -18, CERC, US Army Engineer Waterway Experiment Station, Vicksburg, MS, 1987

Goda, Y. ; Random seas and design of maritime structures, University of Tokyo Press, 1985

Hotta, S. and Mizuguchi, M; A field study of waves in the surf zone, Coastal Engineering in Japan, Vol. 23, 1980, pp. 59-79

Longuet-Higgins, M. S. The statistical analysis of a random, moving surface, Phil. Trans. Roy. Soc. London, Ser. A (1966), Vol. 249, 1957, pp. 321 - 387.

Smith, J. M. and Kraus, N. C.; SUPERTANK laboratorydata collection project, Volume II, Appendices A-I, Technical Report CERC-94-3, Waterway Experiment Station, US Army Corps of Engineers, 1995

Wiegel, R. L. and Saville, T. Jr.; History and heritage of coastal engineering, ASCE, 1996, pp. 513-600

Appendix: Program listing for WAVETRAP

```
C      DATE  April. 1 '97
C *****************************************
C **                                      *
C **  SUBPROGRAM FOR puv data ANALYSIS    *
C **  THIS SUBPROGRAM USES AN INDIVIDUAL  *
C **  WAVE APPROACH. PROGRAM GIVES YOU    *
C **  MAXIMUM, SIGNIFICANT, MEAN WAVE     *
C **  CONDITION AND PROBABILITY.          *
C **             (April 1997)             *
C **   Please note; This is the limitted  *
C **   version of the developed program.  *
C **        Ryuichiro Nishi               *
C **                                      *
C *****************************************

C       PROGRAM WTES1
C
C      TOP
      PARAMETER (K1=14000)
      PARAMETER (K2=5000)
      DIMENSION X(0:K1),PXR(0:100),DXR(0:100),NU(0:K2),ND(0:K2)
      DIMENSION ETU(0:K2),ETD(0:K2),HU(0:K2),HD(0:K2)
      DIMENSION TU(0:K2),TD(0:K2),HSU(0:K2),HSD(0:K2),ETL(0:1)
      DIMENSION XSORT(0:K2),ASORT(0:K2)
      DIMENSION H(0:K2),T(0:K2),HS(0:50),TS(0:50),PH(0:50),PT(0:50)
      DIMENSION JPHT(0:50,0:50)
      DIMENSION SIWEHO(0:K1)
      COMPLEX CO(0:K1), XXX(0:K1)
      CHARACTER FNAME1*45

      COMMON /BLK1/ XSORT, ASORT

      WRITE (*,*) '=============================================='
      WRITE (*,*) '            WAVE DATA ANALYSIS PROGRAMME      '
      WRITE (*,*) '                                              '
      WRITE (*,*) '        (STATISTICS OF INDIVIDUAL WAVES)      '
      WRITE (*,*) '                                              '
      WRITE (*,*) '=============================================='
      WRITE (*,*) ' '
C      DO 24 I=1, 100000

C  24  CONTINUE
      WRITE (*,*) '            Definition of individual wave     '
C
      WRITE (*,*) '+n(t)                                          '
      WRITE (*,*) ' I          zero-up cross method              '
      WRITE (*,*) ' I    I<----------------------------->I        '
      WRITE (*,*) ' I    I                              I         '
      WRITE (*,*) ' I    I        * *                   I         '
      WRITE (*,*) ' I    I     *        *               I    * *  '
      WRITE (*,*) ' I    I  *             *             I  *    * '
      WRITE (*,*) ' I    I*                 *           I*       *'
      WRITE (*,*) '-I---*---------------------*-------------*----------*->t'
      WRITE (*,*) 'OI *                       I *         *       I* '
      WRITE (*,*) ' I*                        I   *     *         I  *'
      WRITE (*,*) ' *                         I      * *          I '
      WRITE (*,*) ' I                         I                   I '
      WRITE (*,*) ' I                         I<----------------->I '
      WRITE (*,*) ' I                           zero-down cross method '
      DO 22 I=1, 100000

   22 CONTINUE

      WRITE (*,*) ' '
      WRITE (*,*) ' '
      WRITE (*,*) '!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!'
      WRITE (*,*) '              PROGRAM  WTES1.FOR               '
```

```
      WRITE (*,*) '               (V.950700CBI)                  '
      WRITE (*,*) '                                              '
      WRITE (*,*) 'IT IS RECOMMENDED TO ADD SOME GRAPHIC SUBROUTINES.'
      WRITE (*,*) ' DATA FILES READY FOR GRAPHIC ARE AS FOLLOWS;  '
      WRITE (*,*) '                                              '
      WRITE (*,*) '   1.Time series of free surface elevation    '
      WRITE (*,*) '   2.TIME SERIES OF WAVE HEIGHT               '
      WRITE (*,*) '   3.TIME SERIES OF WAVE PERIOD               '
      WRITE (*,*) '   4.TIME SERIES OF MEAN WATER LEVE           '
      WRITE (*,*) '    (mean of water surface elevation          '
      WRITE (*,*) '           for each individual wave)          '
      WRITE (*,*) '   5.PROBABILITY OF WATER SURFACE ELEVATION   '
      WRITE (*,*) '   6.PROBABILITY OF WAVE HEIGHT               '
      WRITE (*,*) '    (Rayleigh distribution can be given)      '
      WRITE (*,*) '   7.PROBABILITY OF WAVE PERIOD               '
      WRITE (*,*) '     (Rayleigh distribution can be given)     '
      WRITE (*,*) '   7. JOINT PROBABILITY DISTRIBUTION OF H AND T  '
      WRITE (*,*) '                                              '
      WRITE (*,*) '!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!'
      WRITE (*,*) ' '
      WRITE (*,*) '  '
      DO 20 I=1, 100000

   20 CONTINUE

      WRITE (*,*) ' ==============================================='
      WRITE (*,*) '          WAVE DATA ANALYSIS PROGRAM           '
      WRITE (*,*) '  THIS PROGRAM GIVES YOU SOME INFORMATION ON;   '
      WRITE (*,*) '  SURFACE ELEVATION, REPRESENTATIVE WAVES, ETC.  '
      WRITE (*,*) '             VERSION 950700CBI                 '
      WRITE (*,*) ' When you need some more information,please     '
      WRITE (*,*) ' contact with Ryuichiro Nishi FAX.099-285-8483  '
      WRITE (*,*) ' ==============================================='
      WRITE (*,*) '  '
      WRITE (*,*) '  '
      WRITE (*,*) '  '

      PI=3.14159
      G = 9.80665

C     ' ///////////// <<   INPUT CONDITION   >> ///////////////

C      ---- TEST PARAMETERS -------
C            FOR A2214A9.DAT
             N = 6000
             DT = 0.05
C            FOR CALIBRATION
             CMC = 1.0
C     --------------------------
      NUMD = N
C     ' READ (*,*) 'PLEASE INPUT THE DATA CODE ',
C     ' WRITE (*,*) 'DATA CODE
C     ' WRITE (*,'(A)') '   '
C     ' READ (*,*) 'PLEASE INPUT THE FILE NAME',
C     ' WRITE (*,'(A)') '  '
C     ' READ ' NUMBER OF THE DATA ', N
C     ' WRITE (*,*) ' NUMBER OF THE DATA IS ',N," POINTS"
C     ' WRITE (*,'(A)') '  '
C     ' READ (*,*) 'PLEASE INPUT THE SAMPLING TIME ', DT
C     ' WRITE (*,*) 'SAMPLING TIME IS',DT, 'sec'
C     ' WRITE (*,'(A)') '  '
C     ' READ (*,*) 'PLEASE INPUT THE CONVERSION FACTOR', CMC
C     ' WRITE (*,*) 'CONVERSION FACTOR IS ', CMC
C     ' WRITE (*,'(A)') ' '

C      ////////////////   DATA INPUT   ////////////////////
```

```
C     DATA INPUT TEMPORARY ONLY FOR SURFACE ELEVATION
      FNAME1 ='¥PCWAVE¥ptest10.dat'
      OPEN (3, FILE= FNAME1 )
C     READ (3,*) INUMDAT, SAMPDT
C     READ (3,*) ICH1, ICH2, ICH3, ICH4, ICH5, ICH16
      DO 100 I=0, N-1
C       READ (3,*)  X(I), DUM2, DUM3, DUM4, DUM5, DUM16
        READ (3,*) WIJ, X(I)
 100  CONTINUE
      CLOSE (3, STATUS='KEEP')

C     DATA CONVERSION
      DO 110 I=0, N-1
        X(I)=CMC*X(I)
 110  CONTINUE

C     DATA MEAN
      XSUM=0.0
      DO 120 I=0, N-1
        XSUM=XSUM+X(I)
  120 CONTINUE
      XMEAN=XSUM/FLOAT(N)
      DO 125 I=0, N-1
        X(I)=X(I)-XMEAN
  125 CONTINUE

C     DATA MAX
      XMAX=0.0
      DO 130 I=0, N-1
        IF (ABS(X(I)).GT.XMAX) XMAX=ABS(X(I))
  130 CONTINUE

C    /////////////////////
C     *HIGH PASS FILTER
      HIPANS=2.0
      IF (HIPANS.EQ.1.0) THEN
        WRITE (*,*) ' PLEASE INPU THE CUT OFF FREQUENCY '
        READ (*,*) CFREQ
        CALL HIPASS(NUMD,CFREQ,DT,X)
      ELSE
        CONTINUE
      END IF
  150 CONTINUE

C    /////////////////////////////////////////////////
C     * -----<< rms ,skewness, kurtosis >>----------
      X1SUM=0.0
      X2SUM=0.0
      X3SUM=0.0
      DO 160 I=0, N-1
        XS=X(I)*X(I)
        X1SUM=X1SUM+XS
        XC=X(I)*XS
        X2SUM=X2SUM+XC
        XQ=X(I)*XC
        X3SUM=X3SUM+XQ
  160 CONTINUE
      RMS =SQRT(X1SUM/FLOAT(N))
      SKEW=X2SUM/(RMS**3*FLOAT(N))
      KURT=X3SUM/(RMS**4*FLOAT(N))

C    '-----<< PROBABILITY DISTRIBUTION OF
C             SURFACE ELEVATION >>------
      DO 170 I=0, N-1
        XR=X(I)/RMS
        J=INT((XR+4.)*4.)
        PXR(J)=PXR(J)+1.0
```

```
  170 CONTINUE
      DO 180 J=0, 31
        PXR(J)=4.*PXR(J)/FLOAT(N)
  180 CONTINUE
      DXR1=0.0
      DO 190 J=0, 15
        DXR2=DXR1+.25
        DXR(J+16)= DXR1
        DXR(J+17)= DXR2
        DXR(16-J)=-DXR1
        DXR(15-J)=-DXR2
        DXR1=DXR2
  190 CONTINUE
C      WRITE (*,*) ' PROBABILITY DISTRIBUTION OF SURFACE ELEVATION '
       CALL PROUT2(DXR,PXR)
C     ////////////////////////////////////////////////
C      '-------<< zero cross analysis >>-------
C           DETERMINATION OF ZERO CROSS POINTS
C     ////////////////////////////////////////////////
      J1=0
      J2=0
      K=N-2
      DO 210 I=0, K
        II=I+1
        IF (X(I)*X(II).LT.0.0 .AND. X(II).GT.0.0) THEN
          NU(J1)=II
          J1=J1+1
        END IF
        IF (X(I)*X(II).LT.0.0 .AND. X(II).LT.0.0) THEN
          ND(J2)=II
          J2=J2+1
        ENDIF
  210 CONTINUE
C     NUMBER OF WAVES
      NWU=J1-1
      NWD=J2-1

C     --------<< ZERO-UP AND -DOWN ANALYSIS >>---------
      TT=FLOAT(N)*DT
      DO 220 I=3, 0, -1
        IF (TT.GE.300.*2.**I) THEN
          ETL(0)=300.*2.**I
          GOTO 222
        END IF
  220 CONTINUE
      ETL(0)=150.0
  222 ETL(1)=2.0*ETL(0)

      NUM=NWU-1
      IND=1
C     ' DO NOT ERSAE THIS NUMBER
  235 CONTINUE
      HMAX=0.0
      TMAX=0.0
      DO 230 I=0, NUM
        I1=I
        I2=I+1
        IF (IND.EQ.1) THEN
          M1=NU(I1)
          M2=NU(I2)
        END IF
        IF (IND.EQ.2) THEN
          M1=ND(I1)
          M2=ND(I2)
        END IF
```

```
          SUMX=0.0
          XMAX=X(M1)
          XMIN=X(M1)
          JMAX=M1
          JMIN=M1
          MM=M2-1
          DO 240 J=M1, MM
            SUMX=SUMX+X(J)
            IF (X(J).GE.XMAX) THEN
              XMAX=X(J)
              JMAX=J
            END IF
            IF (X(J).LE.XMIN) THEN
              XMIN=X(J)
              JMIN=J
            END IF
  240     CONTINUE
          IF (IND.EQ.1) THEN
            HSU(I)=SUMX/FLOAT(M2-M1+1)
            ETU(I)=FLOAT(M1-1)*DT+FLOAT(M2-M1)*DT/2.0
          END IF
          IF (IND.EQ.2) THEN
            HSD(I)=SUMX/FLOAT(M2-M1+1)
            ETD(I)=FLOAT(M1-1)*DT+FLOAT(M2-M1)*DT/2.0
          END IF
            A=(X(JMAX-1)-2.*X(JMAX)+X(JMAX+1))*.5
            B=(X(JMAX+1)-X(JMAX-1))*.5
            C=X(JMAX)
            D=(X(JMIN-1)-2.*X(JMIN)+X(JMIN+1))*.5
            E=(X(JMIN+1)-X(JMIN-1))*.5
            F=X(JMIN)
          IF (IND.EQ.1) THEN
C           ZERO UP-CROSS
            HU(I)=C-B**2/(4.*A)-(F-E**2/(4.*D))
            IF (HU(I).GT.HMAX) HMAX=HU(I)
            TU(I)=FLOAT(M2-M1)*DT+(X(M1)/(X(M1)-X(M1-1))-X(M2)/
     &            (X(M2)-X(M2-1)))*DT
            IF (TU(I).GT.TMAX) TMAX=TU(I)
          END IF
          IF (IND.EQ.2) THEN
C           ZERO DOWN-CROSS
            HD(I)=C-B**2/(4.*A)-(F-E**2/(4.*D))
            IF (HD(I).GT.HMAX) HMAX=HD(I)
            TD(I)=FLOAT(M2-M1)*DT+(X(M1)/(X(M1)-X(M1-1))-X(M2)/
     &            (X(M2)-X(M2-1)))*DT
            IF (TD(I).GT.TMAX) TMAX=TD(I)
          END IF

  230 CONTINUE

C       '---  PRINT OUT OF INDIVIDUAL WAVE CHARACTERISTICS  ---
        CALL PROUT3(NUM,HU,TU,HSU,HD,TD,HSD)

C        --- SWITCH TO ZERO-DOWN CROSS METHOD  ---
        IF (IND.EQ.2) GO TO 270
        NUM=NWD-1
        IND=2
        GO TO 235
C
  270 CONTINUE

C       Hrms for zero down and zero-up
        SSS1 = 0.0
        DO 280 I=1, NWU
          SSS1 = SSS1 + HU(I)**2
  280 CONTINUE
        HRMSU = SQRT(SSS1 / FLOAT(NWU))
```

```
      SSS2 = 0.0
      DO 282 I=1, NWD
        SSS2 = SSS2 + HD(I)**2
 282  CONTINUE
      HRMSD = SQRT(SSS2 / FLOAT(NWD))

C     ///////////////////////////////////////////////////////
C     '----<< COMP. OF REPRESENTATIVE WAVE HEIGHT >>--------
C     ///////////////////////////////////////////////////////
      DO 300 IND=1, 2
        IF (IND.EQ.1) THEN
          NN=NWU-1
        ELSE
          NN=NWD-1
        END IF
        IF (IND.EQ.2) THEN
          GO TO 293
        END IF
        DO 290 I=0, NN
          XSORT(I)=HU(I)
          ASORT(I)=TU(I)
  290   CONTINUE
        GOTO 292
  293   DO 295 I=0, NN
          XSORT(I)=HD(I)
          ASORT(I)=TD(I)
  295   CONTINUE
  292   CONTINUE

        CALL PSORT(NN)

        IF (IND.EQ.2) GO TO 296
        DO 297 I=0, NN
          HU(I)=XSORT(I)
          TU(I)=ASORT(I)
  297   CONTINUE
        GOTO 299
  296   DO 298 I=0, NN
          HD(I)=XSORT(I)
          TD(I)=ASORT(I)
  298   CONTINUE
  299   CONTINUE
  300 CONTINUE

C '  ***  COMP. OF REPRESENTATIVE WAVE HEIGHTS ***

 2610 HMAXU=HU(0)
      TMAXU=TU(0)
      HMAXD=HD(0)
      TMAXD=TD(0)
      HSUMU=0.0
      TSUMU=0.0
      HSUMD=0.0
      TSUMD=0.0
      MU1=INT(NWU/10)
      MD1=INT(NWD/10)
      MU2=INT(NWU/3)
      MD2=INT(NWD/3)
      DO 310 I=0, MU1-1
        HSUMU=HSUMU+HU(I)
        TSUMU=TSUMU+TU(I)
  310 CONTINUE
      DO 320 I=0, MD1-1
        HSUMD=HSUMD+HD(I)
        TSUMD=TSUMD+TD(I)
  320 CONTINUE
C     H1/10
```

```
      H10U=HSUMU/MU1
      T10U=TSUMU/MU1
      H10D=HSUMD/MD1
      T10D=TSUMD/MD1
      DO 330 I=MU1, MU2-1
        HSUMU=HSUMU+HU(I)
        TSUMU=TSUMU+TU(I)
  330 CONTINUE
      DO 340 I=MD1, MD2-1
        HSUMD=HSUMD+HD(I)
        TSUMD=TSUMD+TD(I)
  340 CONTINUE
C     H1/3
      H3U=HSUMU/MU2
      T3U=TSUMU/MU2
      H3D=HSUMD/MD2
      T3D=TSUMD/MD2
      DO 350 I=MU2, NWU-1
        HSUMU=HSUMU+HU(I)
        TSUMU=TSUMU+TU(I)
  350 CONTINUE
      DO 360 I=MD2, NWD-1
        HSUMD=HSUMD+HD(I)
        TSUMD=TSUMD+TD(I)
  360 CONTINUE
C     Hmean
      HMEANU=HSUMU/NWU
      TMEANU=TSUMU/NWU
      HMEAND=HSUMD/NWD
      TMEAND=TSUMD/NWD

C    ////////////////////////////////////////////////
C    '--<< PROBABILITY DISTRIBUTION OF H AND T >>----
C    ////////////////////////////////////////////////

      DO 400 IND=1, 2
       IF (IND.EQ.1) THEN
         N=NWU-1
        ELSE
         N=NWD-1
        END IF
C        WRITE (*,*) ' ***  DISTRIBUTIONS OF H AND T  *** '
        IF (IND.EQ.2)  GO TO 415
        DO 410 I=0, N
          H(I)=HU(I)/HMEANU
          T(I)=TU(I)/TMEANU
  410   CONTINUE
C        WRITE (*,*) '== zero-up cross method ==  ### waves', NWU
        GOTO 425
  415   CONTINUE
        DO 420 I=0, N
          H(I)=HD(I)/HMEAND
          T(I)=TD(I)/TMEAND
  420   CONTINUE
C       WRITE (*,*) '== zero-down cross method ==  ### waves', NWD
  425   CONTINUE
C       WRITE (*,*) '  '
C       WRITE (*,*) 'H/Hmean   Prob(H/Hmean)   T/Tmean   Prob(T/Tmean)'
        DO 430 J=0, N
          II=INT(H(J)*4.0)
          PH(II)=PH(II)+1.0
          JJ=INT(T(J)*4.0)
          PT(JJ)=PT(JJ)+1.0
          JPHT(II,JJ)=JPHT(II,JJ)+1
  430   CONTINUE
        IF (IND.EQ.1) THEN
          WRITE (*,*) ' zero-up cross method '
```

```
          WRITE (*,*) 'NUMBER OF WAVES = ', NWU
          WRITE (*,*) 'Hmax    =  '  ,HMAXU
          WRITE (*,*) 'H(1/10)=  '   ,H10U
          WRITE (*,*) 'H(1/3) =  '   ,H3U
          WRITE (*,*) 'Hrmsup =  '   ,HRMSU
          WRITE (*,*) 'Hmean  =  '   ,HMEANU
          WRITE (*,*) 'Tmax   =  '   ,TMAXU
          WRITE (*,*) 'T(1/10)=  '   ,T10U
          WRITE (*,*) 'T(1/3) =  '   ,T3U
          WRITE (*,*) 'Tmean  =  '   ,TMEANU
          WRITE (*,*) 'See the "READMEWS.FOR" for possible graphic data'
        ELSE
          WRITE (*,*) ' zero-down cross method '
          WRITE (*,*) 'NUMBER OF WAVES = ', NWD
          WRITE (*,*) 'Hmax    =  '  ,HMAXD
          WRITE (*,*) 'H(1/10)=  '   ,H10D
          WRITE (*,*) 'H(1/3) =  '   ,H3D
          WRITE (*,*) 'Hrmsdown =  '   ,HRMSD
          WRITE (*,*) 'Hmean  =  '   ,HMEAND
          WRITE (*,*) 'Tmax   =  '   ,TMAXD
          WRITE (*,*) 'T(1/10)=  '   ,T10D
          WRITE (*,*) 'T(1/3) =  '   ,T3D
          WRITE (*,*) 'Tmean  =  '   ,TMEAND
          WRITE (*,*) 'See the "READMEWS.FOR" for possible graphic data'
        END IF

        CALL PROUT4(IND,N,HS,TS,PH,PT)
        CALL RAYLEI
        CALL PROUT5(IND,JPHT)

C     ////////////////////////////////////////////////
C      WRITE (*,*) ' JOINT DISTRIBUTION OF H AND T '
C     ////////////////////////////////////////////////

      OPEN (4,FILE='JHT1.DAT')
      DO 450 I=16, 1, -1
        XP = FLOAT(I)*0.25
        DO 455 J=0, 16
          YP = FLOAT(J)*0.25
             WRITE (4,*) XP, YP, JPHT(I,J)
  455   CONTINUE
  450 CONTINUE
      CLOSE (4, STATUS='KEEP')

      SUMHT=0.0
      SUMH2=0.0
      SUMT2=0.0
      DO 460 I=0, N
        SUMHT=SUMHT+H(I)*T(I)
        SUMH2=SUMH2+H(I)**2
        SUMT2=SUMT2+T(I)**2
  460 CONTINUE
      SOKAN=(SUMHT-FLOAT(N)-1.)/(SQRT(SUMH2-FLOAT(N)-1.)
     &      *SQRT(SUMT2-FLOAT(N)-1.))
C      WRITE (*,*) ' AUTO-CORRELATION           = ',SOKAN

  400 CONTINUE

      CALL PROUT1(FNAME1,NUMD,DT,RMS,SKEW,KURT,NWU,NWD,HMAXU,H10U,
     &  H3U,HMEANU,TMAXU,T10U,T3U,TMEANU,HMAXD,H10D,H3D,HMEAND,TMAXD,
     &  T10D,T3D,TMEAND,X, HRMSU, HRMSD)

C             ^^^^^^^^^^^^^^^^^^^^^^^^
C              PROGRAM IS TERMINATED
C             ------------------------

       STOP
```

```
      END

C       +++++++++++++++++++++++++++++++++++++++
C       ++                                  ++
C       ++    SUBROUTINES DATED Sept. 8 '94  ++
C       ++    OTHER SUBROUTINES EXPECTED TO  ++
C       ++    ADDED MIGHT BE;                ++
C       ++     1. PEAK TO PEAK METHOD        ++
C       ++     2. WEIBLE DISTRIBUTION        ++
C       ++     3. SPECTRUM ANALYSIS          ++
C       ++     4. MAXIMUM ENTROPY METHOD     ++
C       ++                                  ++
C       +++++++++++++++++++++++++++++++++++++++

C     ////////////////////////////////////////////////////
C     '-----------<< HIGH-PASS FILTER >>---------------
C     ////////////////////////////////////////////////////
C
      SUBROUTINE HIPASS(NUMD,CFREQ,DT,X)
      PARAMETER (K1=14000)
      DIMENSION X(0:K1)

      PI = 3.14159
      AF=2*PI*CFREQ*DT
      AAF=1-.5*AF
      A1F=1-AF
      X0=X(0)
      X(0)=AAF*X(0)
      DO 600 NIF=1, NUMD-1
        XI=X(NIF)
        X(NIF)=A1F*X(NIF)+AAF*(XI-X0)
        X0=XI
  600 CONTINUE
      RETURN
      END

C    ////////////////////////////////////////////////////
C    '-----------<< SORTING PROGRAM >>---------------
C    ////////////////////////////////////////////////////
C
      SUBROUTINE PSORT(NN)
      PARAMETER (K2=5000)
      DIMENSION XSORT(0:K2), ASORT(0:K2)
      DIMENSION  LSORT(0:K2),IRSORT(0:K2)
      COMMON /BLK1/ XSORT, ASORT

      KST = 0
      LST = 0
      IRST = NN

  650  IF (LST .GT. IRST)  GO TO 670
       IST= LST
       JST= IRST + 1
       TST= XSORT(LST)
       SST= ASORT(LST)
  655  JST= JST-1
       IF  (IST .GE. JST)  GO TO  665
       IF  (TST .GE. XSORT(JST)) GO TO 655
       XSORT(IST) = XSORT(JST)
       ASORT(IST) = ASORT(JST)
  660  IST= IST+1
       IF (IST .GE. JST)  GO TO 665
       IF (XSORT(IST) .GE. TST)  GO TO 660
       XSORT(JST) = XSORT(IST)
       ASORT(JST) = ASORT(IST)
```

```
      GOTO 655
  665 XSORT(JST) = TST
      ASORT(JST) = SST
      KST = KST+1
      LSORT(KST) = JST+1
      IRSORT(KST) = IRST
      IRST = JST-1
      GOTO 650
  670 IF (KST.EQ.0)  GO TO 675
      LST = LSORT(KST)
      IRST = IRSORT(KST)
      KST = KST-1
      GOTO 650
  675 CONTINUE

      RETURN
      END

C     //////////////////////////////
C        PRINT OUT  AND OUTPUT
C     //////////////////////////////

C     +++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
      SUBROUTINE PROUT1(FNAME1,NUMD,DT,RMS,SKEW,KURT,NWU,NWD,HMAXU,H10U,
     &  H3U,HMEANU,TMAXU,T10U,T3U,TMEANU,HMAXD,H10D,H3D,HMEAND,TMAXD,
     &   T10D,T3D,TMEAND,X, HRMSU,HRMSD)
C     +++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

      PARAMETER (K1=14000)
      DIMENSION X(0:K1)
      CHARACTER FNAME1*12

C     ----------- RAW DATA ------------
      OPEN (5, FILE='ELEV2.DAT')
      DO 140 I=0, NUMD-1
         WRITE (5,*) FLOAT(I)*DT, X(I)
  140 CONTINUE
      CLOSE (5, STATUS='KEEP')
C
C     ---------- REPRESENTATIVE WAVE CONDITION  -----------
      OPEN (6, FILE='WSINFO10.DAT')
        WRITE (6,*) ' WAVE INFORMATION BY ZERO CROSS METHOD'
        WRITE (6,*) ' DATA ANALYSIS BY V950700CBI '
        WRITE (6,*) ' WAVE DATA      = '     ,FNAME1
        WRITE (6,*) ' number of data = '           , NUMD
        WRITE (6,*) ' sumpling time  = '           , DT, ' sec'
        WRITE (6,*) '  '
        WRITE (6,*) ' r.m.s n(t) = ', RMS
        WRITE (6,*) ' skewnwss    = ', SKEW
        WRITE (6,*) ' kurutosis   = ', KURT


C     ' -- output of rep. wave conditions ----
C     '  zero-up cross RESULT

        WRITE (6,*) '  '
        WRITE (6,*) ' zero-up cross method '
        WRITE (6,*) 'NUMBER OF WAVES = ', NWU
        WRITE (6,*) 'Hmax    =  '  ,HMAXU
        WRITE (6,*) 'H(1/10)=  '   ,H10U
        WRITE (6,*) 'H(1/3) =  '   ,H3U
        WRITE (6,*) 'Hrmsup =  '   ,HRMSU
        WRITE (6,*) 'Hmean   =  '  ,HMEANU
        WRITE (6,*) 'Tmax    =  '  ,TMAXU
        WRITE (6,*) 'T(1/10)=  '   ,T10U
        WRITE (6,*) 'T(1/3) =  '   ,T3U
```

```
        WRITE (6,*) 'Tmean   = '  ,TMEANU
        WRITE (6,*) '  '
        WRITE (6,*) ' zero-down cross method '
        WRITE (6,*) 'NUMBER OF WAVES = ', NWD
        WRITE (6,*) 'Hmax    = '   ,HMAXD
        WRITE (6,*) 'H(1/10)= '    ,H10D
        WRITE (6,*) 'H(1/3) = '    ,H3D
        WRITE (6,*) 'Hmean  = '    ,HMEAND
        WRITE (6,*) 'Hrmsdown = ' ,HRMSD
        WRITE (6,*) 'Tmax    = '   ,TMAXD
        WRITE (6,*) 'T(1/10)= '    ,T10D
        WRITE (6,*) 'T(1/3) = '    ,T3D
        WRITE (6,*) 'Tmean  = '    ,TMEAND
        WRITE (6,*) '  '
        WRITE (6,*) '  --> Please see the "READMEWS.FOR" '
        WRITE (6,*) '      regarding output <-- '

      CLOSE (6, STATUS='KEEP')
      RETURN
      END

C      ++++++++++++++++++++++++++++++++
          SUBROUTINE PROUT2(DXR,PXR)
C      ++++++++++++++++++++++++++++++++

      DIMENSION DXR(0:100), PXR(0:100)

C      WRITE (*,*) ' n/n rms    Prob( n/n rms) '
       OPEN (6, FILE='PBETA1.DAT')
       DO 200 I=0, 31
         DXX2 = (DXR(I)+DXR(I+1))/2.0
         WRITE (6,*) DXX2, PXR(I)
  200  CONTINUE
       CLOSE (6, STATUS='KEEP')
       RETURN
       END

C     ++++++++++++++++++++++++++++++++++++++++++++++++
         SUBROUTINE PROUT3(NUM,HU,TU,HSU,HD,TD,HSD)
C     ++++++++++++++++++++++++++++++++++++++++++++++++

      PARAMETER (K2=4000)
      DIMENSION HU(0:K2),HD(0:K2),TU(0:K2),TD(0:K2),
     &          HSU(0:K2),HSD(0:K2)

C      WRITE (*,*) ' TIME SERIES OF H, T, AND M.W.L. '
C      WRITE (*,*) ' ==  zero-up cross method  == '
C      WRITE (*,*) ' ==  zero-down cross method  == '

      OPEN (2, FILE='THSERI1U.DAT')
      OPEN (3, FILE='TTSERI1U.DAT')
      OPEN (4, FILE='TMSERI1U.DAT')
      OPEN (5, FILE='THSERI1D.DAT')
      OPEN (6, FILE='TTSERI1D.DAT')
      OPEN (7, FILE='TMSERI1D.DAT')

      DO 260 J0=0, NUM
          WRITE (2, *) J0, HU(J0)
          WRITE (3, *) J0, TU(J0)
          WRITE (4, *) J0, HSU(J0)
          WRITE (5, *) J0, HD(J0)
          WRITE (6, *) J0, TD(J0)
          WRITE (7, *) J0, HSD(J0)
  260 CONTINUE
      CLOSE (7, STATUS='KEEP')
      CLOSE (6, STATUS='KEEP')
      CLOSE (5, STATUS='KEEP')
```

```
      CLOSE (4, STATUS='KEEP')
      CLOSE (3, STATUS='KEEP')
      CLOSE (2, STATUS='KEEP')
      RETURN
      END

C     +++++++++++++++++++++++++++++++++++++++++
         SUBROUTINE PROUT4(IND,N,HS,TS,PH,PT)
C     +++++++++++++++++++++++++++++++++++++++++

      DIMENSION HS(0:50),TS(0:50),PH(0:50),PT(0:50)
      CHARACTER FNAME2*10, FNAME3*10

      HS(0)=0.0
      TS(0)=0.0
      IF (IND.EQ.1) THEN
        FNAME2 = 'PH1U.DAT'
        FNAME3 = 'PT1U.DAT'
      ELSE
        FNAME2 = 'PH1D.DAT'
        FNAME3 = 'PT1D.DAT'
      END IF

      OPEN (4, FILE=FNAME2 )
      OPEN (5, FILE=FNAME3 )
      DO 100  I=0, 16
        HS(I+1)=HS(I)+.25
        TS(I+1)=TS(I)+.25
        PH(I)=4.*PH(I)/(N+1)
        PT(I)=4.*PT(I)/(N+1)
        HS2 = (HS(I)+HS(I+1))/2.0
        TS2 = (TS(I)+TS(I+1))/2.0
        WRITE (4,*) HS2, PH(I)
        WRITE (5,*) TS2, PT(I)
  100 CONTINUE
      CLOSE (5, STATUS='KEEP')
      CLOSE (4, STATUS='KEEP')
      RETURN
      END

C     /////////////////////////
         SUBROUTINE RAYLEI
C     /////////////////////////
C     ** RAYLEIGH DISTRIBUTION

C     ' WAVE HEIGHT DIST.
C
      OPEN (6, FILE='RAYH1.DAT' )
      HV=3.14159/4.
      DO 30 I=1, 35
        RX=FLOAT (I)/ 10.0
        RY =2*HV*RX*EXP(-HV*RX*RX)
        WRITE (6,*) RX, RY
   30 CONTINUE
      CLOSE (6, STATUS='KEEP')
C
C     ' FREQUENCY DIST.
      OPEN (6, FILE='RAYT1.DAT')
      DO 35 I=1, 35
        RX = FLOAT(I)/10.0
        TV = 0.675*RX**3
        RY = 4.0*TV*EXP(-TV*RX)
        WRITE (6,*) RX, RY
   35 CONTINUE
      CLOSE (6, STATUS='KEEP')
      RETURN
```

```
      END

C     ////////////////////////////////////
         SUBROUTINE PROUT5(IND,JPHT)
C     ////////////////////////////////////

      DIMENSION JPHT(0:50,0:50)
      CHARACTER FNAME2*11

C      WRITE (*,*) ' JOINT DISTRIBUTION OF H AND T '

C     WRITE (*,*) 'H/Hmean '
      IF (IND.EQ.1) THEN
        FNAME2 = 'JHT1U.DAT'
      ELSE
        FNAME2 = 'JHT1D.DAT'
      END IF

      OPEN (4,FILE=FNAME2 )
      DO 450 I=16, 1, -1
        XP = FLOAT(I)*0.25
        DO 455 J=0, 16
          YP = FLOAT(J)*0.25
          WRITE (4,*) XP, YP, JPHT(I,J)
  455   CONTINUE
  450 CONTINUE
      CLOSE (4, STATUS='KEEP')
      RETURN
      END


C       NOTE; Spectrum analysis routines have been
C             deleted (April 1997).
C      ***********************************
C           NO MORE PROGRAMING
C
C           ********        NN      N
C           *       *       N N     N
C           *       *       N  N    N
C           ********        N   N   N
C           *      *        N    N  N
C           *       *       N     N N
C           *         * @   N      NN @
C
C           Free information is available from
C
C           Ryuichiro Nishi
C           Department of Ocean Civil Engineering
C           Kagoshima University
C           1-21-40 Korimoto, Kagoshima
C           JAPAN
C           Tel. +81-992-85-8483  Fax. +81-992-85-8483
C           sediment@oce.eng.kagoshima-u.ac.jp
C
C
C     ****************************************
```